Deflection and buckling of simply supported deep beam

This example considers the basic case of a simply supported beam with uniformly distributed load as shown in Figure 1. However, the analysis is non-trivial because of the axial force, *P*, and the fact that the cross-section height, *h*, is large compared with the length, *L*. That means that this is a deep beam. It is assumed made of glulam with *N* laminates of thickness 38mm and width *b*. The example is also spiced up by employing a collection of different analysis methods. The main objective is to determine the midspan deflection.



Figure 1: The beam considered in this example.

Input values in kN and m

```
val1 = {N → 46};
val2 = {E → 13 100.0 × 10^3, ν → 0.0, b → 0.215, L → 8.0, q → 100.0,
    P → 10 000};
val3 = {h → 0.038 N, G → E / (2 (1 + ν^2))};
val4 = {I → b h^3 / 12, Av → 5/6 b h};
```

Deflection from formula sheet

Most collections of formulas for structural analysis contains the deflection of a simply supported beam with uniformly distributed load. By substituting the given input values and multiplying by 10^3 to obtain the answer in mm we obtain:

$$\text{wFormulaMidspan} = 10^3 \frac{5\, L^4\, q}{384\, E\, I} \text{ /. val4 /. val3 /. val2 /. val1}$$

which yields: 4.25447

Deflection by solving the differential equation

The differential equation for Euler-Bernoulli beam theory contains the fourth-order derivative of the lateral displacment *w*:

$$\text{ddddwExact} = -\frac{q}{E\, I};$$

To obtain the general solution to this differential equation we integrate it four times. First integration:

```
dddwExact = Integrate[ddddwExact, x] + C1
```

which yields: $C1 - \frac{q\, x}{E\, I}$

Second integration:

```
ddwExact = Integrate[dddwExact, x] + C2
```

which yields: $C2 + C1\, x - \frac{q\, x^2}{2\, E\, I}$

Third integration:

```
dwExact = Integrate[ddwExact, x] + C3
```

which yields: $C3 + C2\, x + \frac{C1\, x^2}{2} - \frac{q\, x^3}{6\, E\, I}$

Fourth integration:

```
wExact = Integrate[dwExact, x] + C4
```

which yields: $C4 + C3\, x + \frac{C2\, x^2}{2} + \frac{C1\, x^3}{6} - \frac{q\, x^4}{24\, E\, I}$

That general solution has four unknown integration constants that is determined by the following four boundary conditions:

- Zero displacement at A: $w(0) = 0$

- Zero bending moment at A: $M(0) = EI\, w'' = 0$
- Zero displacement at B: $w(L) = 0$
- Zero bending moment at B: $M(L) = EIw'' = 0$

Those four boundary conditions are formulated in the following four equations, using the displacement expression established earlier:

```
Eq1 = wExact /. x → 0;
Eq2 = E I ddwExact /. x → 0;
Eq3 = wExact /. x → L;
Eq4 = E I ddwExact /. x → L;
```

Solving those four equations in the four unknowns yields:

```
soln = Solve[{Eq1 == 0, Eq2 == 0, Eq3 == 0, Eq4 == 0}, {C1, C2, C3, C4}] //
  Simplify
```

which yields: $\left\{\left\{C1 \to \frac{L\, q}{2\, E\, I},\ C2 \to 0,\ C3 \to -\frac{L^3\, q}{24\, E\, I},\ C4 \to 0\right\}\right\}$

That gives the following expression for the displacement of the simply supported beam:

```
wExact = wExact /. soln
```

which yields: $\left\{-\frac{L^3\, q\, x}{24\, E\, I} + \frac{L\, q\, x^3}{12\, E\, I} - \frac{q\, x^4}{24\, E\, I}\right\}$

That expression is here plotted for the given input values:

```
Plot[wExact /. val4 /. val3 /. val2 /. val1, {x, 0, L /. val2},
 PlotStyle → Black]
```



Symbolically, the displacement at midspan is recognized from many formula sheets:

$$wExactMidspan = wExact\ /.\ x \to \frac{L}{2}$$

which yields: $\left\{-\frac{5 L^4 q}{384 E I}\right\}$

That maximum displacment can be written in another way by evaluating the fraction:

```
wExactMidspan // N
```

which yields: $\left\{-\frac{0.0130208 L^4 q}{E I}\right\}$

Numerically the downwards displacement at midspan is, in mm:

$$\Delta exact = 10^3\ wExactMidspan\ /.\ val4\ /.\ val3\ /.\ val2\ /.\ val1$$

which yields: {−4.25447}

Deflection by virtual work

This method enables us to determine the shear deformation in addition to the flexural (bending) deformation that we normally consider. In the virtual work approach we place a unit point load at midspan to obtain the displacement at that location. The peak values for the virtual bending moment diagram (BMD) and shear force diagram (SFD) are:

$$\text{Mvirtual} = \frac{1\,L}{4};$$

$$\text{Vvirtual} = \frac{1}{2};$$

The real BMD and SFD, due to the distributed load, has the following peak values:

$$\text{Mreal} = \frac{q\,L^2}{8};$$

$$\text{Vreal} = \frac{q\,L}{2};$$

Integration yields the following flexural deformation:

$$\triangle\text{flex} = \frac{5}{12\,E\,I}\,\text{Mvirtual}\,\text{Mreal}\,L$$

which yields: $\frac{5\,L^4\,q}{384\,E\,I}$

That means the flexural deformation is, in mm:

```
△flexNumerical = 10^3 △flex /. val4 /. val3 /. val2 /. val1
```

which yields: 4.25447

Similarly, integration yields the following shear deformation:

$$\triangle\text{shear} = \frac{1}{2\,G\,Av}\,\text{Vvirtual}\,\text{Vreal}\,L$$

which yields: $\frac{L^2\,q}{8\,Av\,G}$

Numerically, the shear deformation is, in mm:

```
△shearNumerical = 10^3 △shear /. val4 /. val3 /. val2 /. val1
```

which yields: 0.389987

The total deflection, including flexure and shear is:

```
ΔflexNumerical + ΔshearNumerical
```

which yields: 4.64446

The shear deformation can be plotted as a fraction of the total deformation for different number of laminates, *N:*

```
Plot[ Δshear 100 / (Δflex + Δshear) /. val4 /. val3 /. val2, {N, 20, 60},
 PlotRange → {{20, 60}, {0, 30}}, GridLines → Automatic,
 AxesLabel → {"N", "Shear percentage"}, PlotStyle → Black]
```



Deflection by stiffness method

To determine the midpoint deflection by the stiffness method we need a degree of freedom (DOF) there. Different choices are possible for the other DOFs and the choice in Figure 2 is made. The figure also introduces the auxiliary length Le, which is the length of each of the two "finite elements:"

```
Le = L/2;
```

Figure 2: Degrees of freedom.

The forces along the DOFs to give a unit displacement along the 1st DOF are, including shear deformations:

$$\text{Kcolumn1} = \frac{1}{(1+\alpha)}\left\{2\,\frac{12\,E\,I}{Le^3},\ \frac{6\,E\,I}{Le^2},\ 0,\ -\frac{6\,E\,I}{Le^2}\right\};$$

The forces along the DOFs to give a unit rotation along the 2nd DOF are:

$$\text{Kcolumn2} = \frac{1}{(1+\alpha)}\left\{\frac{6\,E\,I}{Le^2},\ \frac{(4+\alpha)\,E\,I}{Le},\ \frac{(2-\alpha)\,E\,I}{Le},\ 0\right\};$$

The forces along the DOFs to give a unit rotation along the 3rd DOF are:

$$\text{Kcolumn3} = \frac{1}{(1+\alpha)}\left\{0,\ \frac{(2-\alpha)\,E\,I}{Le},\ 2\,\frac{(4+\alpha)\,E\,I}{Le},\ \frac{(2-\alpha)\,E\,I}{Le}\right\};$$

The forces along the DOFs to give a unit rotation along the 4th DOF are:

$$\text{Kcolumn4} = \frac{1}{(1+\alpha)}\left\{-\frac{6\,E\,I}{Le^2},\ 0,\ \frac{(2-\alpha)\,E\,I}{Le},\ \frac{(4+\alpha)\,E\,I}{Le}\right\};$$

Bringing together the columns of the stiffness matrix into one matrix we obtain:

```
Ksymb = {Kcolumn1, Kcolumn2, Kcolumn3, Kcolumn4};
```

Next we evaluate the constant α for shear deformation:

$$\text{alpha} = \frac{12\,E\,I}{G\,Av\,Le^2}\ /.\ \text{val4}\ /.\ \text{val3}\ /.\ \text{val2}\ /.\ \text{val1}$$

which yields: 0.458326

That gives the following values in the stiffness matrix:

```
K = Ksymb /. α → alpha /. val4 /. val3 /. val2 /. val1;
K // MatrixForm
```

which yields:

$$\begin{pmatrix} 322351. & 322351. & 0 & -322351. \\ 322351. & 958098. & 331307. & 0 \\ 0 & 331307. & 1.9162\times10^{6} & 331307. \\ -322351. & 0 & 331307. & 958098. \end{pmatrix}$$

Next we assemble the "clamping force vector," namely the forces to keep the DOFs clamped under the imposed load:

$$\text{Ftilde} = \left\{2\,\frac{q\,Le}{2},\ -\frac{q\,Le^2}{12},\ 0,\ \frac{q\,Le^2}{12}\right\};$$

The actual load vector is the negative of the clamping force vector:

```
F = -Ftilde /. val4 /. val3 /. val2 /. val1;
```

That gives the following values in the load vector:

```
MatrixForm[F]
```

which yields: $\begin{pmatrix} -400. \\ 133.333 \\ 0 \\ -133.333 \end{pmatrix}$

Solving the system of equilibrium equations yields the following displacements and rotation along the DOFs:

```
u = Inverse[K].F
```

which yields: $\{-0.00464446,\ 0.00170179,\ -7.11508 \times 10^{-20},\ -0.00170179\}$

It is understood that a more efficient way of doing that is:

```
LinearSolve[K, F]
```

which yields: $\{-0.00464446,\ 0.00170179,\ -3.43193 \times 10^{-20},\ -0.00170179\}$

The first DOF is the displacement at midspan with both flexural and shear deformation. That value in mm is:

$$u[[1]]\ 10^3$$

which yields: -4.64446

Deflection by energy and polynomial shape

In the energy method we assume a deformed shape and use the principle of minimum potential energy to determine the amplitude of that displacement. Here we try with the polynomial shape function:

$$\text{wPoly} = \left(\frac{4}{L^2} x^2 - \frac{4}{L} x\right) \triangle\text{Poly};$$

```
Plot[wPoly /. val2 /. △Poly → 1, {x, 0, L /. val2}, PlotStyle → Black]
```



The shape function is differentiated twice for use in the expression for strain energy:

```
dwPoly = D[wPoly, x]
```

which yields: $\left(-\frac{4}{L}+\frac{8\,x}{L^2}\right)\triangle\mathrm{Poly}$

```
ddwPoly = D[dwPoly, x]
```

which yields: $\frac{8\,\triangle\mathrm{Poly}}{L^2}$

The strain energy is:

$$\mathrm{strainEnergy} = \int_0^L \frac{1}{2}\,E\,I\,\mathrm{ddwPoly}^2\,dx$$

which yields: $\frac{32\,\triangle\mathrm{Poly}^2\,E\,I}{L^3}$

The potential energy in the applied load is:

$$\mathrm{loadPotential} = \int_0^L -q\,\mathrm{wPoly}\,dx$$

which yields: $\frac{2\,L\,q\,\triangle\mathrm{Poly}}{3}$

The total potential energy is:

```
Π = strainEnergy - loadPotential;
```

We set the derivative of the total potential energy equation to zero to determine the displacement amplitude:

```
equation = D[Π, ΔPoly];
solnPoly = Solve[equation == 0, ΔPoly];
solnPoly /. val4 /. val3 /. val2 /. val1
```

which yields: `{{ΔPoly → 0.00340358}}`

From the displacement we can determine the bending moment at midspan, here expressing the answer as the factor to be compared with the number 8 in the classic formula $\frac{qL^2}{8}$:

$$\frac{L^2\, q}{E\, I\, \text{ddwPoly}} \;/.\; \left\{\text{solnPoly}[[1, 1]],\ x \to \frac{L}{2}\right\} \;//\; N$$

which yields: `12.`

We observe that the displacement result was perhaps somewhat accurate but the bending moment not.

Deflection by energy and trigonometric shape

The trigonometric shape function explored here is:

$$\text{wTrig} = -\Delta\text{Trig}\ \text{Sin}\left[\frac{\pi}{L}\, x\right];$$

```
Plot[wTrig /. val2 /. ΔTrig -> 1, {x, 0, L /. val2}, PlotStyle → Black]
```



The shape function is next differentiated and substituted into the strain energy integral:

```
dwTrig = D[wTrig, x]
```

which yields: $-\frac{\pi\,\Delta\text{Trig}\,\text{Cos}\left[\frac{\pi\,x}{L}\right]}{L}$

```
ddwTrig = D[dwTrig, x]
```

which yields: $\frac{\pi^2\,\Delta\text{Trig}\,\text{Sin}\left[\frac{\pi\,x}{L}\right]}{L^2}$

$$\text{strainEnergy} = \int_0^L \frac{1}{2}\,E\,I\,\text{ddwTrig}^2\,dx$$

which yields: $\frac{\pi^4\,\Delta\text{Trig}^2\,E\,I}{4\,L^3}$

The potential energy of the load must also be recalculated because the shape function is different now:

$$\text{loadPotential} = \int_0^L -q\,\text{wTrig}\,dx$$

which yields: $\frac{2\,L\,q\,\Delta\text{Trig}}{\pi}$

The maximum displacement is now:

```
Π = strainEnergy - loadPotential;
equation = D[Π, △Trig];
solnTrig = Solve[equation == 0, △Trig];
solnTrig /. val4 /. val3 /. val2 /. val1
```

which yields: `{{△Trig → 0.00427088}}`

That displacement is quite accurate; the used shape function is a good approximation of the exact solution. The bending moment at midspan is again calculated as a factor to be compared with 8 in $\frac{qL^2}{8}$:

$$\frac{L^2 q}{E\,I\,\text{ddwTrig}} \;/.\; \left\{\text{solnTrig}[[1,1]],\; x \to \frac{L}{2}\right\} \;//\; N$$

which yields: `7.75157`

It may seem puzzling that the displacement from the energy method with a trigonometric function gives a slightly HIGHER displacement, i.e., seemingly softer behaviour, compared with the exact solution. However, while the approximate trigonometric shape is higher at midspan it is lower elsewhere. Here is the difference between the exact and the energy solution along the beam:

```
lineExact = wExact /. val4 /. val3 /. val2 /. val1;
lineEnergy = wTrig /. solnTrig /. val4 /. val3 /. val2 /. val1;
Plot[{ lineEnergy - lineExact}, {x, 0, L /. val2}, PlotStyle → Black]
```



Here is the displaced shape near the end:

```
Plot[{lineEnergy, lineExact}, {x, 0.2 L /. val2, 0.22 L /. val2},
 PlotStyle → {{Black}, {Black, Dashed}},
 PlotLegends → Placed[{"Trigonometric shape", "Exact shape"}, Below]]
```



Here is the displaced shape at midspan, showing that the approximate solution dips further down here:

```
Plot[{lineEnergy, lineExact}, {x, 0.48 L /. val2, 0.52 L /. val2},
 PlotStyle → {{Black}, {Black, Dashed}},
 PlotLegends → Placed[{"Trigonometric shape", "Exact shape"}, Below]]
```



Deflection with P-delta

Select stiffness coefficients in the geometric stiffness matrix similar to what was done for the elastic stiffness matrix:

$$\text{KGcolumn1} = \left\{2\,\frac{6\,P}{5\,Le},\ \frac{P}{10},\ 0,\ -\frac{P}{10}\right\};$$

$$\text{KGcolumn2} = \left\{\frac{P}{10},\ \frac{2\,P\,Le}{15},\ -\frac{P\,Le}{30},\ 0\right\};$$

$$\text{KGcolumn3} = \left\{0,\ -\frac{P\,Le}{30},\ 2\,\frac{2\,P\,Le}{15},\ -\frac{P\,Le}{30}\right\};$$

$$\text{KGcolumn4} = \left\{-\frac{P}{10},\ 0,\ -\frac{P\,Le}{30},\ \frac{2\,P\,Le}{15}\right\};$$

Bring together the columns into one matrix:

```
KG = {KGcolumn1, KGcolumn2, KGcolumn3, KGcolumn4};
```

Calculate the deflection with P-delta but without shear deformation:

```
Ktotal = (K - KG) /. α → 0 /. val4 /. val3 /. val2 /. val1;
Ktotal // MatrixForm
```

which yields:
$$\begin{pmatrix} 316351. & 321351. & 0 & -321351. \\ 321351. & 952765. & 332641. & 0 \\ 0 & 332641. & 1.90553\times 10^6 & 332641. \\ -321351. & 0 & 332641. & 952765. \end{pmatrix}$$

Solve system of equations:

```
utotal = Inverse[Ktotal].F
```

which yields: {-0.00492015, 0.00179943, 0., -0.00179943}

The displacement at midspan in mm is now larger due to the P-delta effect:

```
utotal[[1]] 10^3
```

which yields: -4.92015

Euler buckling load

When we now turn to buckling due to axial force, i.e., NOT lateral torsional buckling, it is understood that the beam will buckle sideways about the weak axis unless it is restrained from moving sideways. Hence, the moment of inertia, *I*, in the following formulas should be interpreted as *I* for bending about the weak axis. The exact Euler buckling load is:

$$\text{PcrEuler} = \frac{\pi^2 \, E\, I}{L^2} \; // \; N$$

which yields: $\frac{9.8696 \, E\, I}{L^2}$

Buckling load from matrix analysis

Because there are four DOFs there are four buckling loads

$$\text{eigenVals} = \text{Eigenvalues}\left[\left\{\text{Ksymb} \;/.\; \alpha \to 0,\; \frac{\text{KG}}{P}\right\}\right] \; // \; N$$

which yields: $\left\{\frac{48.\, E\, I}{L^2}, \frac{240.\, E\, I}{L^2}, \frac{9.94385\, E\, I}{L^2}, \frac{128.723\, E\, I}{L^2}\right\}$

The lowest eigenvalue governs the buckling load. As seen, that result is slightly higher than the exact Euler buckling load because polynomial shape functions are used to derive the geometric stiffness coefficients used above, while the exact solution involves trigonometric functions.

Buckling load from energy and polynomial shape

Total potential energy:

$$\Pi = \int_0^L \frac{1}{2}\, E\, I\, \text{ddwPoly}^2\, dx - \int_0^L \frac{1}{2}\, P\, \text{dwPoly}^2\, dx$$

which yields: $-\frac{8\, P\, \Delta\text{Poly}^2}{3\, L} + \frac{32\, \Delta\text{Poly}^2\, E\, I}{L^3}$

Differentiate Π and set it equal to zero:

```
equation = D[Π, △Poly];
Solve[equation == 0, P]
```

which yields: $\left\{\left\{P \to \frac{12\, E\, I}{L^2}\right\}\right\}$

That result is higher than the exact Euler buckling load because the exact solution involves trigonometric functions.

Buckling load from energy and trigonometric shape

Total potential energy:

$$\Pi = \int_0^L \frac{1}{2}\, E\, I\, \text{ddwTrig}^2\, dx - \int_0^L \frac{1}{2}\, P\, \text{dwTrig}^2\, dx$$

which yields: $-\frac{P\, \pi^2\, \Delta\text{Trig}^2}{4\, L} + \frac{\pi^4\, \Delta\text{Trig}^2\, E\, I}{4\, L^3}$

Differentiate Π and set it equal to zero:

```
equation = D[Π, △Trig];
Solve[equation == 0, P]
```

which yields: $\left\{\left\{P \to \frac{\pi^2\, E\, I}{L^2}\right\}\right\}$

This result is exact because the exact solution involves trigonometric functions.